Some remarks on Solar eclipses

A practical calculation method for the eclipse obscuration taking solar limb darkening into account

Robert Nufer

12. Mai 2017

The progress of a solar eclipse is described by either the eclipse magnitude, which is the ratio of the Sun's diameter occulted by the Moon, or the eclipse obscuration, which is the fraction of the Sun's surface occulted by the Moon. Typically, the eclipse obscuration is calculated in a purely geometrically manner, neglecting Solar limb darkening. Mathematical formulae can be found in the Explanatory Supplement to the Astronomical Almanac. An interactive online version is provided by Adrian Jannetta.

I have prepared an Excel workbook for the calculation of the Eclipse obscuration, and the ***"Eclipse dimming"***, which takes solar limb darkening into account.

The Excel file can be downloaded here.



Figure 1: Eclipse obscuration is the proportion of the area C G C' F C compared to the Sun's disk area. In this example, 12.20 % of the Sun's disk area are obscured by the Moon.

For the calculation of the (geometrical) eclipse obscuration, the areas of segment C-D-C'-F-C and C-G-C'-D-C have to be calculated.

For the calculation of the ***"Eclipse dimming"***, a different approach is used: Instead of computing disk segments, the Sun is divided into a number (for example *n*=100) of concentric annuli from the Sun's center to the Sun's limb at radius 1. The width of each annulus in units of the solar radius is 1/n. The surface brightness for each annulus is defined by a limb darkening function. In a first attempt the practical formula with only one parameter for the green light at wave length 540 nm with β = 2, found in Wikipedia is used:

$$\frac{I_r}{I_0} = \frac{1+\beta\sqrt{1-r^2}}{1+\beta} , \qquad \text{(ED)}$$

where *r* is the distance to the Sun's center or the radius of the annulus.

The parameter β defines the amount of limb darkening. In the visible wavelength range, β can be set between about 0.9 (800 nm) and 10 (380 nm). If β = 0, no limb darkening is assumed, and the result is identical to the geometrical Eclipse obscuration formula.



Figure 2: The black curve shows the assumed limb darkening using β = 2.0 in formula (ED) (540 nm; green). The red curve shows the contribution to the surface brightness for the annuli. The innermost annuli have highest surface brightness, but due to their small size, contribution to total brightness is small. On the other side, the outermost annuli are not the ones with maximum contribution to total brightness. Using β = 2.0, maximum contributing annuli is found at ~0.81 solar radii.



Figure 3: The function Eclipse_dimming computes the sector occulted by the Moon for all *n* annuli, where function (ED) is used. In this example, the outermost annulus at *r* = 1 is shown. From this annulus, sector C C' is occulted by the Moon. From annulus at *r* = 0.75 sector D D' is occulted by the Moon. Compared to the geometrical Eclipse obscuration (black curve and point 12.20%), ***"Eclipse dimming"*** effect is smaller (10.66%) in this example, because, during early and late phase of the solar eclipse, only "darker" zones of the Sun's disk are then occulted. At maximum eclipse, ***"Eclipse dimming"*** effect is maximal, because the brightest central zones are occulted. For the same reason, ***"Eclipse dimming"*** effect during annular phase is not constant as is the case for the simple geometrical Eclipse obscuration.

Basic Listing for the calculation of the *"Eclipse dimming" (ED)*:

```
'===============================================================================
'Eclipse_Dimming computes the dimming in percent of the total solar brightness.
'Arguments Xmoon and Rmoon in units of solar radii
'Xmoon : position of the Moon's center
'Rmoon : Apparent size of the Moon
'n     : number of annuli
'beta  : center limb variation factor (2=540nm; green)
'===============================================================================

Function Eclipse_dimming(XMoon As Double, RMoon As Double, n As Long, beta As
Double) As Double
Dim TotBrightness As Double
Dim ActBrightness As Double
Dim AnnulusRadius As Double
Dim AnnulusBright As Double
Dim BetaSun       As Double
Dim tmp1          As Double
Dim tmp2          As Double
Dim i             As Long
Dim TotalCovered  As Boolean

TotBrightness = 0#
ActBrightness = 0#
For i = n To 1 Step -1
  AnnulusRadius = i / n - 1# / (2# * n)
  tmp1 = XMoon ^ 2# + AnnulusRadius ^ 2# - RMoon ^ 2#
  tmp2 = 2# * Abs(XMoon) * AnnulusRadius
  If (Abs(tmp1) > tmp2) Or tmp2 = 0# Then BetaSun = 0# Else BetaSun = Applica-
tion.Acos(tmp1 / tmp2) * RD
  AnnulusBright = (1# + beta * Sqr(1# - AnnulusRadius ^ 2#)) / (1# + beta) *
AnnulusRadius
  TotBrightness = TotBrightness + AnnulusBright
  TotalCovered = (XMoon + RMoon >= AnnulusRadius) And (XMoon - RMoon <= -
AnnulusRadius)
  If TotalCovered Then ActBrightness = ActBrightness + AnnulusBright _
                  Else ActBrightness = ActBrightness + AnnulusBright * BetaSun /
180#
Next i
Eclipse_dimming = 100# * ActBrightness / TotBrightness
End Function
```

An alternative formula for the center to limb darkening was proposed by SEML member Glenn Schneider:

$$F(u) = 1 - a\left(1 - u^{b}\right), with\ u = \sqrt{1 - r^2} \qquad \text{(ED2)}$$

With parameters a=0.85; b=0.80, as published in [Schneider et al. 2006 in AJ 641 565](https://iopscience.iop.org/article/10.1086/500776/fulltext/).



Figure 4: The blue curve shows the assumed limb darkening using a=0.85 and b=0.80 in formula (ED2). The red curve shows the contribution to the surface brightness for the annuli. The innermost annuli have highest surface brightness, but due to their small size, contribution to total brightness is small. On the other side, the outermost annuli are not the ones with maximum contribution to total brightness. Using the given values for parameters a and b, maximum contributing annuli is found at ~0.78 solar radii.



Figure 5: The function Eclipse_dimming_2 computes the sector occulted by the Moon for all *n* annuli. In this example, the outermost annulus at *r* = 1 is shown. From this annulus, sector C C' is occulted by the Moon. From annulus at *r* = 0.75 sector D D' is occulted by the Moon. Compared to the geometrical Eclipse obscuration (black curve and point 12.20%), ***"Eclipse dimming"*** effect is smaller (10.42%) in this example, because, during early and late phase of the solar eclipse, only "darker" zones of the Sun's disk are then occulted. At maximum eclipse, ***"Eclipse dimming"*** effect is maximal, because the brightest central zones are occulted. For the same reason, ***"Eclipse dimming"*** effect during annular phase is not constant as is the case for the simple geometrical Eclipse obscuration.

Basic Listing for the alternative calculation method of the ***"Eclipse dimming"*** *(ED2)*:

```
'==============================================================================
'Eclipse_Dimming_2 computes the dimming in percent of the total solar bright-
ness.
'Arguments in units of solar radii
'Xmoon : position of the Moon's center
'Rmoon : Apparent size of the Moon
'n     : number of annuli
'a,b   : center limb variation factors provided by Glenn Schneider
'        see Hestroffer and Magan (1998) A&A 333 342
(http://aa.springer.de/papers/8333001/2300338.pdf)
'        and Schneider et al. 2006 in AJ 641 565
(http://iopscience.iop.org/article/10.1086/500427/pdf)
'==============================================================================

Function Eclipse_dimming_2(XMoon As Double, RMoon As Double, n As Long, a As
Double, b As Double) As Double
Dim TotBrightness As Double
Dim ActBrightness As Double
Dim AnnulusRadius As Double
Dim AnnulusBright As Double
Dim BetaSun       As Double
Dim tmp1          As Double
Dim tmp2          As Double
Dim i             As Long
Dim TotalCovered  As Boolean

TotBrightness = 0#
ActBrightness = 0#
For i = n To 1 Step -1
  AnnulusRadius = i / n - 1# / (2# * n)
  tmp1 = XMoon ^ 2# + AnnulusRadius ^ 2# - RMoon ^ 2#
  tmp2 = 2# * Abs(XMoon) * AnnulusRadius
  If (Abs(tmp1) > tmp2) Or tmp2 = 0# Then BetaSun = 0# Else BetaSun = Applica-
tion.Acos(tmp1 / tmp2) * RD
  AnnulusBright = (1# - a * (1 - Sqr(1 - AnnulusRadius ^ 2) ^ b)) * AnnulusRadi-
us
  TotBrightness = TotBrightness + AnnulusBright
  TotalCovered = (XMoon + RMoon >= AnnulusRadius) And (XMoon - RMoon <= -
AnnulusRadius)
  If TotalCovered Then ActBrightness = ActBrightness + AnnulusBright _
                  Else ActBrightness = ActBrightness + AnnulusBright * BetaSun /
180#
Next i
Eclipse_dimming_2 = 100# * ActBrightness / TotBrightness
End Function
```

On a first view the difference between the two brightness profiles (ED) and (ED2) is relatively small, but near the Sun's limb it becomes significant. The difference between the profiles of function (ED) (with β = 2.0) and (ED2) (with a = 0.85; b = 0.80) is shown in Figure 6. In this figure the radial brightness of function (ED) is scaled to an identical total solar surface brightness. The inner regions of the Sun between 0 and ~0.7 *r* are slightly brighter using function (ED2), but the outer regions are much darker in (ED2).



Figure 6: Radial solar surface brightness profiles of functions (ED) and (ED2) (with β=2.0) and (ED2) (with a=0.85; b=0.80).

In Figure 7 the difference between the two functions (ED) and (ED2) during a fictive annular solar eclipse of magnitude 0.95 is shown. First contact occurs, when the Moon's center is at -1.95 *r* (solar radii) from the Sun's center. Only small parts of the outermost regions, where the difference between (ED) and (ED2) is relatively large, are then occulted by the Moon during the early eclipse phase. Because the dimming effect of function (ED) is higher in the outer zones of the Sun compared to function (ED2), the difference (ED)-(ED2) increases, until about ten percent of the Sun's disk are occulted by the Moon. At the Moon's distance ~1.08 *r*, the dimming effect of both functions is the same. But from this moment, function (ED2) clearly dominates over (ED), thus the difference is more and more negative until eclipse maximum.

At this moment, the Moon is surrounded by the outermost radial zones of the Sun, which are significantly darker using function (ED2), thus a difference of ~-0.9% in eclipse dimming. In other words, this corresponds to a difference of ~15% in the remaining sunlight at eclipse maximum.



Figure 7: ***"Eclipse dimming"*** for a hypothetical annular solar eclipse with magnitude 0.95. Note the behavior of the red curve (difference (ED)-(ED2)) in the vicinity of the eclipse maximum (right end of the curves). For total eclipses the difference is zero, but for annular eclipses the difference drops to a negative local minimum value.

The differences in **“Eclipse dimming”** between the functions (ED) (with β=2) and (ED2) (with a=0.85; b=0.80) are shown in Figure 8. Eclipse magnitudes between 0.95 (annular eclipses) and 1.05 (total eclipses) are investigated and drawn along the z axis.

For all eclipse magnitudes the difference (ED)-(ED2) is positive at the beginning. Then it reaches a maximum, if the eclipse obscuration is about ten percent. At about thirty percent obstruction the difference is zero again. From this moment, the brighter central regions of the Sun using function (ED2) dominate over (ED), and the difference gets more and more negative.

For total eclipses the difference is then again zero at C2. This is the small horizontal triangle near the right end of Figure 8.

A qualitative difference between function (ED) and (ED2) is observed only for annular eclipses, where the Moon is surrounded by the outermost regions of the Sun. At mid-eclipse the effect is maximal, and it is more pronounced with smaller eclipse magnitudes. This explains the fold on the right edge of the surface graph.



Figure 8: Differences in ***“Eclipse dimming”*** for a series of solar eclipses with magnitudes between 0.95 (near edge; as described before) and 1.05. The distance between the Moon’s center and the Sun’s center is shown on the x axis between shortly before first contact and eclipse maximum.